
Count cells that do not contain many strings

Generic formula 

{ = SUM(1 - (MMULT( --
(ISNUMBER(SEARCH(TRANSPOSE(exclude),data))),ROW(exclude) ^ 0) > 0))}

Summary 

To count cells that do not contain many different strings, you can use a rather complex formula
based on the MMULT function. In the example shown, the formula in F5 is:

{ = SUM(1 - (MMULT( --
(ISNUMBER(SEARCH(TRANSPOSE(exclude),data))),ROW(exclude) ^ 0) > 0))}

where "data" is the named range B5:B14, and "exclude" is the named range D5:D7.

Note: this is an array formula and must be entered with control + shift + enter

Preface

This formula is complicated by the "contains" requirement. If you just need a formula to count
cells that do not *equal* many things, you can use a more straightforward formula based on the
MATCH function. Also, if you have a limited number of strings to exclude, you can use the
COUNTIFS function like this:

= COUNTIFS(data,"<>*pink*",data,"<>*orange*",data,"<>*black*")

However, with this approach, you must enter a new pair of range/criteria arguments for each
string to exclude. In contrast, the formula explained below can handle a large number of strings to
exclude entered directly on the worksheet.

Finally, this formula is complex. Let me know if you have a simpler formula to propose :)

Explanation 

The core of this formula is ISNUMBER and SEARCH:

ISNUMBER(SEARCH(TRANSPOSE(exclude),data))

Here, we transpose the items in the named range "exclude", then feed the result to SEARCH as the
"find text", with "data" as "within text". The SEARCH function returns a 2d array of TRUE and
FALSE values, 10 rows by 3 columns, like this:

{3,#VALUE!,12;#VALUE!,4,#VALUE!;#VALUE!,#VALUE!,#VALUE!;#VALUE!,#VAL
UE!,#VALUE!;#VALUE!,#VALUE!,3;14,#VALUE!,#VALUE!;#VALUE!,#VALUE!,#VA
LUE!;#VALUE!,#VALUE!,#VALUE!;#VALUE!,#VALUE!,#VALUE!;3,#VALUE!,12}

For each value in "data", we have 3 results (one per search string) that are either #VALUE errors or
numbers. Numbers represent the position of a found text string, and errors represent text strings
not found. By the way, the TRANSPOSE function is needed to generate the 10 x 3 array of
complete results.

This array is fed into ISNUMBER to get TRUE FALSE values, which we convert to 1s and 0s with a
double negative (--) operator. The result is an array like this:

{1,0,1;0,1,0;0,0,0;0,0,0;0,0,1;1,0,0;0,0,0;0,0,0;0,0,0;1,0,1}

which goes into the MMULT function as array1. Following the rules of matrix multiplication,
number of columns in array1 must equal the number of rows in array2. To generate array2, we
use the ROW function like this:

ROW(exclude) ^ 0

This yields an array of 1s, 3 rows by 1 column:

{1;1;1}

which goes into MMULT as array2. After array multiplication, we have an array dimensioned to
match the original data:

{2;1;0;0;1;1;0;0;0;2}

In this array, any non-zero number represents a value where at least one of the excluded strings
has been found. Zeros indicate no excluded strings were found. To force all non-zero values to 1,
we use greater than zero:

{2;1;0;0;1;1;0;0;0;2} > 0

which creates yet another array or TRUE and FALSE values:

{TRUE;TRUE;FALSE;FALSE;TRUE;TRUE;FALSE;FALSE;FALSE;TRUE}

Our final goal is to count only text values where no excluded strings were found, so we need to
reverse these values. We do this by subtracting the array from 1. This is an example of boolean
logic. The math operation automatically coerces TRUE and FALSE values to 1s and 0s, and we
finally have an array to return to the SUM function: 

= SUM({0;0;1;1;0;0;1;1;1;0})

The SUM function returns a final result of 5.

Author 

Dave Bruns

Related formulas 

Count cells not equal to many things
First, a little context. Normally, if you have just a couple things
you don't want to count, you can use COUNTIFS like this: =
COUNTIFS ( range , "<>apple" , range , "<>orange" ) But this
doesn't scale very well if...

Count cells not equal to
In Excel, the operator for not equal is " ". For example: = A1 <> 10
// A1 is not equal to 10 = A1 <> "apple" // A1 is not equal to "apple"
The COUNTIF function counts the number of cells in a range that
meet supplied...

Count cells not equal to x or y
The COUNTIFS function counts the number of cells in a range
that meet one or more supplied criteria. All conditions must pass
in order for a cell to be counted. In the example shown, there is a
list of colors in column B in a named range called rng...

Count multiple criteria with NOT logic
The first expression inside of SUMPRODUCTS tests values in
column C, Gender, against the value in F4, "Male": ( gender = F4 )
The result is an array of TRUE FALSE values like this:
{TRUE;FALSE;TRUE;FALSE;TRUE;TRUE;FALSE;TRUE;FALSE}
Where TRUE...

Related functions 

Excel ISNUMBER Function
The Excel ISNUMBER function returns TRUE when a cell contains
a number, and FALSE if not. You can use ISNUMBER to check that
a cell contains a numeric value, or that the result of another
function is a number.

Excel SEARCH Function
The Excel SEARCH function returns the location of one text string
inside another. SEARCH returns the position of the first character
of find_text inside within_text. Unlike FIND, SEARCH allows
wildcards, and is not case-sensitive.

Excel MMULT Function
The Excel MMULT function returns the matrix product of two
arrays. The result from MMULT is an array that contains the same
number of rows as array1 and the same number of columns as
array2. 

Excel TRANSPOSE Function
The Excel TRANSPOSE function "flips" the orientation of a given
range or array: TRANSPOSE flips a vertical range to a horizontal
range, and flips a horizontal range to a vertical range.

Excel Formula Training
Formulas are the key to getting things done in
Excel. In this accelerated training, you'll learn how to
use formulas to manipulate text, work with dates and
times, lookup values with VLOOKUP and INDEX &
MATCH, count and sum with criteria, dynamically
rank values, and create dynamic ranges. You'll also
learn how to troubleshoot, trace errors, and fix
problems. Instant access. See details here.

Hi - I'm Dave Bruns, and
I run Exceljet with my
wife, Lisa. Our goal is to
help you work faster in

Excel. We create short videos, and clear
examples of formulas, functions, pivot
tables, conditional formatting, and
charts. Read more.

I saved hours of time in randomly assigning students to work groups each week by following the
instructions in the above article, along with my class list. Kudos—and thank you! -John (Professor
of Law)

